Computer Algebra, week 1, lecture 3: Rational functions in Maple.

Example:

```
> f:=(x^6-3*x^2+x^5-3*x+x^4-3)/(x^5-3*x^2+x^4-3*x+x^3-3);
```

$$f := \frac{x^6 - 3\,x^2 + x^5 - 3\,x + x^4 - 3}{x^5 - 3\,x^2 + x^4 - 3\,x + x^3 - 3}$$

```
> numer(f); # numerator
```

$$x^6 - 3\,x^2 + x^5 - 3\,x + x^4 - 3$$

```
> denom(f); # denominator
```

$$x^5 - 3\,x^2 + x^4 - 3\,x + x^3 - 3$$

```
> g:=normal(f); # remove gcd of numerator and denominator
```

$$g := \frac{x^4 - 3}{x^3 - 3}$$

The command normal puts rational functions in their normal form, which means the form A/B where A and B are polynomials with no common factors, so the gcd(numer(g), denom(g)) will be 1. It also produces this normal form when you have a sum of rational functions.

```
> f;
```

$$\frac{x^6 - 3\,x^2 + x^5 - 3\,x + x^4 - 3}{x^5 - 3\,x^2 + x^4 - 3\,x + x^3 - 3}$$

```
> f-12*x/(x-3)^3;
```

$$\frac{x^6 - 3\,x^2 + x^5 - 3\,x + x^4 - 3}{x^5 - 3\,x^2 + x^4 - 3\,x + x^3 - 3} - 12\,\frac{x}{(x-3)^3}$$

```
> normal(%);
```

$$\frac{x^7 - 9\,x^6 + 27\,x^5 - 39\,x^4 - 3\,x^3 + 27\,x^2 - 45\,x + 81}{(x-3)^3\,(x^3-3)}$$

```
> f - g;
```

$$\frac{x^6 - 3\,x^2 + x^5 - 3\,x + x^4 - 3}{x^5 - 3\,x^2 + x^4 - 3\,x + x^3 - 3} - \frac{x^4 - 3}{x^3 - 3}$$

```
> normal(%);
```

$$0$$

To test if a rational function is 0, we also use normal.

```
> sqrfree(f);  # Note: this gives an error in Maple 5, but not in 6
  or 7.
Error, (in sqrfree) argument must be a polynomial in, {x}
```

As you can see, a positive multiplicity means that the factor is in the numerator, and negative means it is in the denominator.

```
> factor(x^2+1/x);
```

$$\frac{(x+1)(x^2-x+1)}{x}$$

```
> factors(x^2+1/x);
Error, (in factors) argument must be a polynomial over an algebraic number field
```

If a rational function has a pole at a point alpha of multiplicity e (so with squarefree you'd see a factor with multiplicity -e), then the derivative has a pole of order e+1 (so if you then do sqrfree you'd see a multiplicity -(e+1)).

```
> f:=3/(x-2)^3+x/(x^2+1)^2+1/x;
```

$$f := 3\,\frac{1}{(x-2)^3} + \frac{x}{(x^2+1)^2} + \frac{1}{x}$$

```
> f:=normal(f);
```

$$f := \frac{18\,x^5 + 43\,x^3 + 15\,x - 26\,x^4 - 30\,x^2 + x^7 - 6\,x^6 - 8}{(x-2)^3\,(x^2+1)^2\,x}$$

```
> diff(f,x);
```

$$\frac{90\,x^4 + 129\,x^2 + 15 - 104\,x^3 - 60\,x + 7\,x^6 - 36\,x^5}{(x-2)^3\,(x^2+1)^2\,x} - 3\,\frac{18\,x^5 + 43\,x^3 + 15\,x - 26\,x^4 - 30\,x^2 + x^7 - 6\,x^6 - 8}{(x-2)^4\,(x^2+1)^2\,x} - 4\,\frac{18\,x^5 + 43\,x^3 + 15\,x - 26\,x^4 - 30\,x^2 + x^7 - 6\,x^6 - 8}{(x-2)^3\,(x^2+1)^3} - \frac{18\,x^5 + 43\,x^3 + 15\,x - 26\,x^4 - 30\,x^2 + x^7 - 6\,x^6 - 8}{(x-2)^3\,(x^2+1)^2\,x^2}$$

```
> normal(%);
```

$$-\frac{16 - 32\,x + 189\,x^6 + 65\,x^2 + 172\,x^4 - 208\,x^5 - 72\,x^3 - 8\,x^9 + 39\,x^8 - 80\,x^7 + x^{10}}{(x-2)^4\,(x^2+1)^3\,x^2}$$

```
> normal(diff(%,x));
```

$$2\,(-32 + 80\,x - 360\,x^6 - 208\,x^2 - 330\,x^4 + 233\,x^5 + 378\,x^3 + 552\,x^9 - 832\,x^8 + 912\,x^7 - 180\,x^{10} + 68\,x^{11} - 10\,x^{12} + x^{13}) \Big/ ((x-2)^5\,(x^2+1)^4\,x^3)$$

```
> normal(diff(%,x));
```

$$-6\,(-192\,x + 2691\,x^6 + 560\,x^2 + 2058\,x^4 - 3116\,x^5 - 1120\,x^3 - 1424\,x^9 - 415\,x^8 - 60\,x^7 + 3376\,x^{10} - 2472\,x^{11} + 1280\,x^{12} - 340\,x^{13} + 64 + 105\,x^{14} - 12\,x^{15} + x^{16}) \Big/ ((x-2)^6\,(x^2+1)^5\,x^4)$$

Because of that, the derivative of a rational function can not have pole order 1, if a derivative of a rational function has a pole, the pole order is at least 2.

```
> f:=1/x^5 + 1/x + 1/(x-2)^2 + 1/(x-3);
```

$$f := \frac{1}{x^5} + \frac{1}{x} + \frac{1}{(x-2)^2} + \frac{1}{x-3}$$

```
> int(f,x);
```

$$-\frac{1}{4}\frac{1}{x^4} + \ln(x) - \frac{1}{x-2} + \ln(x-3)$$

Poles of order 1 can not come from derivatives of rational functions. Such poles lead to logarithms when you integrate.

```
> normal(f);
```

$$\frac{x^3 - 7x^2 + 16x - 12 + 2x^7 - 10x^6 + 17x^5 - 12x^4}{x^5 (x-2)^2 (x-3)}$$

```
> g:=normal(f,expanded);
```

$$g := \frac{x^3 - 7x^2 + 16x - 12 + 2x^7 - 10x^6 + 17x^5 - 12x^4}{x^8 - 7x^7 + 16x^6 - 12x^5}$$

Now f was easy to integrate because it was a sum in which each term had the form (...)/(x-alpha)^e. The function g is the same rational function, but looks more complicated. Maple can convert between these forms, see the help page ?convert,parfrac for more information:

```
> convert(g,parfrac,x);
```

$$\frac{1}{x^5} + \frac{1}{x} + \frac{1}{(x-2)^2} + \frac{1}{x-3}$$

```
> g;
```

$$\frac{x^3 - 7x^2 + 16x - 12 + 2x^7 - 10x^6 + 17x^5 - 12x^4}{x^8 - 7x^7 + 16x^6 - 12x^5}$$

To integrate this rational function, Maple will first compute the form given by convert(g,parfrac,x), and after that integration has become much easier:

```
> int(g,x);
```

$$-\frac{1}{4}\frac{1}{x^4} + \ln(x) - \frac{1}{x-2} + \ln(x-3)$$

```
>
```